ZASTOSOWANIA MATEMATYKI
APPLICATIONES MATHEMATICAE
XV, 4 (1977)

ALGORITHM 50

ANNA TRZMIELAK (Wrocław)

JOINING OF CHEBYSHEV SERIES

1. Procedure declaration. Given the Chebyshev series coefficients of the functions g and h, the procedure *ghCheb* calculates the coefficients of the expansion of function f defined by the formula

$$(1)\qquad f(x) = \begin{cases} g(x) & \text{for } -1 \leqslant x \leqslant \xi, \\ h(x) & \text{for } \xi \leqslant x \leqslant 1 \end{cases}$$

into the Chebyshev series of first kind

$$(2)\qquad f(x) = {\sum_{j=0}^{M}}' a_j[f]T_j(x).$$

Data:

N — number of coefficients in the expansions of functions g and h into the Chebyshev series;

M — number of sought coefficients of the Chebyshev series expansion of function f;

ksi — real number ξ occurring in (1);

$g, h[0:N]$ — arrays containing the coefficients of the expansions of functions g and h, respectively.

Results:

$a[0:M]$ — array containing the coefficients of the expansion of function f.

2. Method used. The method of calculation of the coefficients $a_j[f]$ requires the uniform convergence of the Chebyshev series of the functions g and h, and the algorithm assumes that the N first Chebyshev coefficients of these functions are known.

From (1) and from the integral definition of the Chebyshev coefficients ([1], p. 107) we obtain

$$\begin{aligned} a_j[f] &= \frac{2}{\pi}\Big[\int_{-1}^{\xi} g(x)T_j(x)(1-x^2)^{-1/2}dx + \int_{\xi}^{1} h(x)T_j(x)(1-x^2)^{-1/2}dx\Big] \\ &= \frac{2}{\pi}\Big[\int_{-1}^{\xi} {\sum_{k=0}^{\infty}}' a_k[g]T_k(x)T_j(x)(1-x^2)^{-1/2}dx + \\ &\qquad + \int_{\xi}^{1} {\sum_{k=0}^{\infty}}' a_k[h]T_k(x)T_j(x)(1-x^2)^{-1/2}dx\Big], \end{aligned}$$

```
procedure ghCheb(N,M,ksi,g,h,a);
 value N,M,ksi;
 integer N,M;
 real ksi;
 array a,g,h;
 begin
  procedure addCh2(k,s,n,m);
   value s,n,m;
   integer k,n,s,m;
   begin
    integer i;
    real den;
    den:=m;
    for i:=k step s until n do
     begin
      b3:=b[k]/den-b2+b1×x2;
      b2:=b1;
      b1:=b3;
      den:=den-1.0
     end i
   end addCh2;
  array b[0:N];
  integer j,k,r;
  real ar,b1,b2,b3,cj,ks2,s,s1,sj,sj1,x2;
  ks2:=sqrt(1.0-ksi×ksi);
  ar:=1.5707963268-arctan(ksi/ks2);
  x2:=2.0×ksi;
  for k:=0 step 1 until N do
   b[k]:=h[k]-g[k];
  cj:=1.0;
```

```
s:=sin(ar);
b1:=b2:=sj1:=.0;
addCh2(N,-1,1,N);
a[0]:=g[0]+(b[0]×ar+2.0×ks2×(b1×x2-b2))/3.1415926536;
r:=if M<N then M else N;
for j:=1 step 1 until r do
 begin
  b1:=b2:=.0;
  addCh2(N,-1,j+1,N+j);
  b3:=b1×x2-b2;
  b2:=b1;
  b1:=b2;
  k:=j-1;
  addCh2(k,-1,0,j+k);
  addCh2(1,1,k,k);
  s1:=b3;
  b1:=b2:=.0;
  addCh2(N,-1,j+1,N-j);
  sj:=sj1×ksi+cj×s;
  cj:=cj×ksi-sj1×s;
  sj1:=sj;
  a[j]:=2.0×(g[j]+(b[j]×(sj×cj/j+ar)+ks2×(s1+b3))/
        3.1415926536)
 end j;
for j:=N+1 step 1 until M do
 begin
  b1:=b2:=.0;
  addCh2(N,-1,0,N+j);
  addCh2(1,1,N,j-1);
  for k:=j-N-1 step -1 until 1 do
```

```
begin
  b3:=b1×x2-b2;
  b1:=b2;
  b2:=b3
end k;
a[j]:=2.0×ks2×b3/3.1415926536
end j
end ghCheb
```

$$(3)\qquad a_j[f] = \frac{2}{\pi}\sum_{k=0}^{\infty}{}' \frac{1}{2}\Big\{a_k[g]\int\limits_{\arccos\xi}^{\pi}[\cos(k-j)t+\cos(k+j)t]\,dt + a_k[h]\int\limits_{0}^{\arccos\xi}[\cos(k-j)t+\cos(k+j)t]\,dt\Big\}.$$

Formula (3) has different forms for finite expansions and depends upon j as follows:

$$(4)\qquad a_0[f] = \frac{2}{\pi}\Big[\frac{\pi}{2}a_0[g]+\frac{1}{2}(a_0[h]-a_0[g])\arccos\xi + (1-\xi^2)^{1/2}\sum_{k=1}^{N}\frac{1}{k}(a_k[h]-a_k[g])\,U_{k-1}(\xi)\Big],$$

$$(5)\qquad a_j[f] = \frac{2}{\pi}\Big\{\pi a_j[g]+(a_j[h]-a_j[g])\Big(\frac{1}{2j}(1-\xi^2)^{1/2}U_{2j-1}(\xi)+\arccos\xi\Big) + (1-\xi^2)^{1/2}\Big[\sum_{k=0}^{j-1}{}'(a_k[h]-a_k[g])\Big(\frac{1}{j-k}U_{j-k-1}(\xi)+\frac{1}{j+k}U_{j+k-1}(\xi)\Big)\Big] + \sum_{k=j+1}^{N}(a_k[h]-a_k[g])\Big(\frac{1}{k-j}U_{k-j-1}(\xi)+\frac{1}{k+j}U_{k+j-1}(\xi)\Big)\Big\}\quad \text{for } 0<j\leqslant N,$$

$$(6)\qquad a_j[f] = \frac{2}{\pi}(1-\xi^2)^{1/2}\sum_{k=0}^{N}{}'(a_k[h]-a_k[g])\Big(\frac{1}{j-k}U_{j-k-1}(\xi) + \frac{1}{j+k}U_{j+k-1}(\xi)\Big)\quad \text{for } N+1\leqslant j\leqslant M,$$

where the Chebyshev polynomials of second kind are denoted by U_l.

In the procedure the coefficients $a_j[f]$ are calculated after formulas (4)-(6) using the Clenshaw method of calculating the finite linear combination of Chebyshev polynomials of second kind ([1], p. 275, A14.4).

If at least one of the Chebyshev series for g and h is infinite, the use of (6) has no sense. The speed of convergence of series (2) depends mainly on the continuity of f at point ξ and on the continuity of its derivatives.

3. Application. The procedure *ghCheb* has been verified on the Odra 1204 computer using the ALGOL 1204 MT compiler.

Example 1. For the function

$$f(x) = \begin{cases} g(x) = \cos\big(1.5\arccos(x)\big) = \dfrac{3}{\pi}\displaystyle\sum_{k=0}^{\infty}\frac{(-1)^k}{k^2-2.25}T_k(x) & \text{for } -1 \leqslant x \leqslant 0.5, \\ h(x) = -\dfrac{3}{\sqrt{3}}x^2 + \dfrac{12}{\sqrt{3}}x^4 & \text{for } 0.5 \leqslant x \leqslant 1 \end{cases}$$

the following results were obtained:

k	$a_k[f]$			
	$N = 5, M = 5$	$N = 6, M = 7$	$N = 9, M = 9$	$N = 12, M = 14$
0	$.11494100_{10}01$	$.11494102_{10}\ 01$	$.11494100_{10}\ 01$	$.11494100_{10}\ 01$
1	$.18698390_{10}01$	$.18698394_{10}\ 01$	$.18698390_{10}\ 01$	$.18698390_{10}\ 01$
2	$.14185760_{10}01$	$.14502910_{10}\ 01$	$.14502906_{10}\ 01$	$.14502906_{10}\ 01$
3	$-.61582879_{10}00$	$-.61047977_{10}\ 00$	$-.61901292_{10}\ 00$	$-.61901292_{10}\ 00$
4	$.67700851_{10}00$	$.67700905_{10}\ 00$	$.67592486_{10}\ 00$	$.67821099_{10}\ 00$
5	$.12228500_{10}00$	$.61142828_{10}-01$	$.61142499_{10}-01$	$.61074858_{10}-01$
6		$.48597308_{10}-01$	$.24298217_{10}-01$	$.24298217_{10}-01$
7		$.24299147_{10}-01$	$.24299149_{10}-01$	$.24299147_{10}-01$
8			$.27605051_{10}-01$	$.27605051_{10}-01$

Example 2. Let

$$f(x) = \begin{cases} g(x) = \dfrac{x}{x^2-c^2} = -\dfrac{4p^2}{1-p^2}\displaystyle\sum_{k=0}^{\infty}{}' p^{2k}T_{2k+1}(x) & \text{for } -1 \leqslant x \leqslant 0, \\ h(x) = -\dfrac{1}{c^2} - \dfrac{1}{c^2}x & \text{for } 0 \leqslant x \leqslant 1, \end{cases}$$

where $p = c - \operatorname{sgn}(c)(c^2-1)^{1/2}$.

For $c = 1.5$ the results obtained were the following:

k	$a_k[f]$			
	$N = 5, M = 5$	$N = 6, M = 7$	$N = 10, M = 12$	$N = 16, M = 14$
0	$-.44675927_{10}\ 00$	$-.44444444_{10}\ 00$	$-.44444444_{10}\ 00$	$-.44444444_{10}\ 00$
1	$-.16886720_{10}\ 01$	$-.16936103_{10}\ 01$	$-.16936103_{10}\ 01$	$-.16936103_{10}\ 01$
2	$.10806760_{10}\ 00$	$.10136563_{10}\ 00$	$.10136563_{10}\ 00$	$.10136563_{10}\ 00$
3	$.60295553_{10}-01$	$.52359000_{10}-01$	$.52359000_{10}-01$	$.52359000_{10}-01$
4	$.11528028_{10}\ 00$	$.10499216_{10}\ 00$	$.10499216_{10}\ 00$	$.10499216_{10}\ 00$
5	$.18818676_{10}\ 00$	$.86821133_{10}-01$	$.86821133_{10}-01$	$.86821133_{10}-01$
6		$.26127922_{10}\ 00$	$.13063961_{10}\ 00$	$.13063961_{10}\ 00$
7		$.13177531_{10}\ 00$	$.12965330_{10}\ 00$	$.12965330_{10}\ 00$
8			$.11009525_{10}\ 00$	$.11009525_{10}\ 00$

For $c = 2.5$ the following results were obtained:

k	$a_k[f]$			
	$N = 5, M = 5$	$N = 10, M = 9$	$N = 12, M = 14$	$N = 17, M = 18$
0	$-.16005502_{10}\ 00$	$-.16000000_{10}\ 00$	$-.16000000_{10}\ 00$	$-.16000000_{10}\ 00$
1	$-.54577986_{10}\ 00$	$-.54589723_{10}\ 00$	$-.54589723_{10}\ 00$	$-.54589723_{10}\ 00$
2	$.95723103_{10}-02$	$.94130185_{10}-02$	$.94130185_{10}-02$	$.94130185_{10}-02$
3	$.63723109_{10}-02$	$.61836759_{10}-02$	$.61836759_{10}-02$	$.61836759_{10}-02$
4	$.99462380_{10}-02$	$.97017111_{10}-02$	$.97017111_{10}-02$	$.97017111_{10}-02$
5	$.18480345_{10}-01$	$.90673268_{10}-02$	$.90673268_{10}-02$	$.90673268_{10}-02$
6		$.12209931_{10}-01$	$.12209931_{10}-01$	$.12209931_{10}-01$
7		$.12221865_{10}-01$	$.12221865_{10}-01$	$.12221865_{10}-01$
8		$.10220141_{10}-01$	$.10220141_{10}-01$	$.10220141_{10}-01$

The calculation time depends upon N and M, for example for $N = 5$ and $M = 5$ the calculation time was 0.51 sec, and for $N = 16$ and $M = 14$ it was equal to 2.76 sec.

Reference

[1] S. Paszkowski, *Zastosowania numeryczne wielomianów i szeregów Czebyszewa*, PWN, Warszawa 1975.

INSTITUTE OF INFORMATICS
UNIVERSITY OF WROCŁAW
50-384 WROCŁAW

Received on 17. 3. 1976

ALGORYTM 50

ANNA TRZMIELAK (Wrocław)

SPAJANIE SZEREGÓW CZEBYSZEWA

STRESZCZENIE

Mając dane współczynniki Czebyszewa funkcji g i h, precedura *ghCheb* oblicza współczynniki rozwinięcia funkcji (1) w szereg Czebyszewa pierwszego rodzaju (2).

Dane:

N — liczba współczynników rozwinięcia funkcji g i h w szereg Czebyszewa pierwszego rodzaju;

M — liczba szukanych współczynników funkcji f;

ksi — liczba ξ występująca w definicji (1) funkcji f;

g, *h*[*0*:*N*] — tablice współczynników rozwinięcia odpowiednio funkcji g i h.

Wyniki:

a[*0*:*M*] — tablica współczynników rozwinięcia funkcji f.

Przytoczono wyniki obliczeń dla kilku przykładów i różnych wartości N i M. Poprawność procedury sprawdzono na m.c. Odra 1204 przy użyciu translatora ALGOL 1204 MT.